

```
1:
2: // convolve the given image with array of weights
3: PImage convolution(PImage img, float[][] wgt, float offset) {
4:   PImage result = createImage(img.width, img.height, RGB);
5:   for (int x=0; x < img.width; x++) {
6:     for (int y=0; y < img.height; y++) {
7:       result.set(x, y, convolvePixel(img, x, y, wgt, offset));
8:     }
9:   }
10:  return result;
11: }
12:
13: // compute convolution result for pixel (x,y)
14: color convolvePixel(PImage img, int x, int y, float[][] wgt, float offset) {
15:   int h = wgt.length;      // number of rows in weight matrix
16:   int w = wgt[0].length;    // number of entries per row of weight matrix
17:   float r, g, b, total;      // total used for normalization
18:   r = g = b = total = 0;
19:
20:   for (int i=0; i < w; i++) {
21:     int newX = x + i - w/2;
22:     if (newX >= 0 && newX < img.width) { // otherwise out of bounds
23:       for (int j=0; j < h; j++) {
24:         int newY = y + j - h/2;
25:         if (newY >= 0 && newY < img.height) { // otherwise out of bounds
26:           color c = img.get(newX, newY);
27:           r += red(c) * wgt[i][j];
28:           g += green(c) * wgt[i][j];
29:           b += blue(c) * wgt[i][j];
30:           total += wgt[i][j];
31:         }
32:       }
33:     }
34:   }
35:
36:   if (total == 0) {
37:     total = 1; // avoid division by zero errors for neutral kernels
38:   }
39:
40:   return color(offset + r/total, offset + g/total, offset + b/total);
41: }
42:
```